Lineare partielle Differentialgleichungen 1. Ordnung

Eine PDGL mit übersichtlicher, aber nicht trivialer Lösung

In[1]:=
```
solu = DSolve[x ∂x u[x, y] + y ∂y u[x, y] == 2 u[x, y] + 1, u[x, y], {x, y}]
```

Out[1]= $\left\{\left\{u[x,y] \to -\frac{1}{2} + x^2\, c_1\left[\frac{y}{x}\right]\right\}\right\}$

Das zugehörige Vektorfeld. Alle Lösungsflächen müssen in jedem Punkt tangential zum Vektorfeld sein.

In[2]:=
```
vectors = VectorPlot3D[{x, y, 2 z + 1}, {x, 0, 1},
  {y, 0, 1}, {z, 0, 1}, AxesLabel → {x, y, z}, BaseStyle → FontSize → 16]
```

Out[2]=



Hier sollen Stereobilder generiert werden!

In[3]:=
```
GraphicsGrid[{{vectors, vectors}}];
```

Die Projektionen der charakteristischen Kurven auf die xy-Ebene sehen so aus:

In[4]:=
```
solyx = DSolve[y'[x] == y[x]/x, y[x], x]
```

Out[4]= $\{\{y[x] \to x\, c_1\}\}$

In[5]:=
```
projchar = Plot[Table[y[x] /. solyx[[1]] /. C[1] → c, {c, 0, 2, 0.1}],
   {x, 0, 1}, PlotRange → {0, 1}, AspectRatio → Automatic]
```

Out[5]=



In[6]:=
```
sol1 = DSolve[{x'[t] == x[t], y'[t] == y[t], z'[t] == 2 z[t] + 1},
    {x[t], y[t], z[t]}, t] // Simplify
```

Out[6]= $\left\{\left\{x[t] \to e^{t}\, c_1,\ y[t] \to e^{t}\, c_2,\ z[t] \to -\frac{1}{2} + e^{2t}\, c_3\right\}\right\}$

Hier kommen die charak. Kurven in voller Gestalt:

In[7]:=
```
curves = Flatten[Table[{x[t], y[t], z[t]} /. sol1 /. {C[1] → c1, C[2] → c2, C[3] → 0.5},
     {c1, -2, 1, 0.2}, {c2, 0, 1, 0.2}], 2];
```

In[8]:=
```
chars = ParametricPlot3D[curves, {t, -2, 3},
  PlotRange → {{0, 1}, {0, 1}, {0, 1}}, AxesLabel → {x, y, z}, BaseStyle → FontSize → 16]
```

Out[8]=